

NAG Library Routine Document

F08FQF (ZHEEVD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

Warning. The specification of the parameters LRWORK and LIWORK changed at Mark 20 in the case where JOB = 'V' and N > 1: the minimum dimension of the array RWORK has been reduced whereas the minimum dimension of the array IWORK has been increased.

1 Purpose

F08FQF (ZHEEVD) computes all the eigenvalues and, optionally, all the eigenvectors of a complex Hermitian matrix. If the eigenvectors are requested, then it uses a divide-and-conquer algorithm to compute eigenvalues and eigenvectors. However, if only eigenvalues are required, then it uses the Pal-Walker-Kahan variant of the *QL* or *QR* algorithm.

2 Specification

```

SUBROUTINE F08FQF(JOB, UPLO, N, A, LDA, W, WORK, LWORK, RWORK, LRWORK,
1                IWORK, LIWORK, INFO)
    INTEGER
1    N, LDA, LWORK, LRWORK, IWORK(max(1,LIWORK)), LIWORK,
    INFO
    double precision    W(*), RWORK(max(1,LRWORK))
    complex*16          A(LDA,*), WORK(max(1,LWORK))
    CHARACTER*1          JOB, UPLO

```

The routine may be called by its LAPACK name *zheevd*.

3 Description

F08FQF (ZHEEVD) computes all the eigenvalues and, optionally, all the eigenvectors of a complex Hermitian matrix A . In other words, it can compute the spectral factorization of A as

$$A = Z\Lambda Z^H,$$

where Λ is a real diagonal matrix whose diagonal elements are the eigenvalues λ_i , and Z is the (complex) unitary matrix whose columns are the eigenvectors z_i . Thus

$$Az_i = \lambda_i z_i, \quad i = 1, 2, \dots, n.$$

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- 1: JOB – CHARACTER*1 *Input*
On entry: indicates whether eigenvectors are computed.
 JOB = 'N'
 Only eigenvalues are computed.

JOB = 'V'

Eigenvalues and eigenvectors are computed.

Constraint: JOB = 'N' or 'V'.

2: UPLO – CHARACTER*1

Input

On entry: indicates whether the upper or lower triangular part of A is stored.

UPLO = 'U'

The upper triangular part of A is stored.

UPLO = 'L'

The lower triangular part of A is stored.

Constraint: UPLO = 'U' or 'L'.

3: N – INTEGER

Input

On entry: n , the order of the matrix A .

Constraint: $N \geq 0$.

4: A(LDA,*) – **complex*16** array

Input/Output

Note: the second dimension of the array A must be at least $\max(1, N)$.

On entry: the n by n Hermitian matrix A .

If UPLO = 'U', the upper triangular part of A must be stored and the elements of the array below the diagonal are not referenced.

If UPLO = 'L', the lower triangular part of A must be stored and the elements of the array above the diagonal are not referenced.

On exit: if JOB = 'V', A is overwritten by the unitary matrix Z which contains the eigenvectors of A .

5: LDA – INTEGER

Input

On entry: the first dimension of the array A as declared in the (sub)program from which F08FQF (ZHEEVD) is called.

Constraint: $LDA \geq \max(1, N)$.

6: W(*) – **double precision** array

Output

Note: the dimension of the array W must be at least $\max(1, N)$.

On exit: the eigenvalues of the matrix A in ascending order.

7: WORK(max(1, LWORK)) – **complex*16** array

Workspace

On exit: if INFO = 0, the real part of WORK(1) contains the required minimal size of LWORK.

8: LWORK – INTEGER

Input

On entry: the dimension of the array WORK as declared in the (sub)program from which F08FQF (ZHEEVD) is called.

If LWORK = -1, a workspace query is assumed; the routine only calculates the minimum dimension of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Constraints:

if $N \leq 1$, $LWORK \geq 1$ or $LWORK = -1$;

if JOB = 'N' and $N > 1$, $LWORK \geq N + 1$ or $LWORK = -1$;

if JOB = 'V' and $N > 1$, $LWORK \geq N \times (N + 2)$ or $LWORK = -1$.

- 9: RWORK(max(1, LRWORK)) – **double precision** array *Workspace*
On exit: if INFO = 0, RWORK(1) contains the required minimal size of LRWORK.
- 10: LRWORK – INTEGER *Input*
On entry: the dimension of the array RWORK as declared in the (sub)program from which F08FQF (ZHEEVD) is called.
 If LRWORK = -1, a workspace query is assumed; the routine only calculates the minimum dimension of the RWORK array, returns this value as the first entry of the RWORK array, and no error message related to LRWORK is issued.
Constraints:
 if $N \leq 1$, $LRWORK \geq 1$ or $LRWORK = -1$;
 if JOB = 'N' and $N > 1$, $LRWORK \geq N$ or $LRWORK = -1$;
 if JOB = 'V' and $N > 1$, $LRWORK \geq 2 \times N^2 + 5 \times N + 1$ or $LRWORK = -1$.
- 11: IWORK(max(1, LIWORK)) – INTEGER array *Workspace*
On exit: if INFO = 0, IWORK(1) contains the required minimal size of LIWORK.
- 12: LIWORK – INTEGER *Input*
On entry: the dimension of the array IWORK as declared in the (sub)program from which F08FQF (ZHEEVD) is called.
 If LIWORK = -1, a workspace query is assumed; the routine only calculates the minimum dimension of the IWORK array, returns this value as the first entry of the IWORK array, and no error message related to LIWORK is issued.
Constraints:
 if $N \leq 1$, $LIWORK \geq 1$ or $LIWORK = -1$;
 if JOB = 'N' and $N > 1$, $LIWORK \geq 1$ or $LIWORK = -1$;
 if JOB = 'V' and $N > 1$, $LIWORK \geq 5 \times N + 3$ or $LIWORK = -1$.
- 13: INFO – INTEGER *Output*
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO = - i , argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

if INFO = i and JOB = 'N', the algorithm failed to converge; i elements of an intermediate tridiagonal form did not converge to zero; if INFO = i and JOB = 'V', then the algorithm failed to compute an eigenvalue while working on the submatrix lying in rows and column $i/(N+1)$ through mod($i, N+1$).

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix ($A + E$), where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the **machine precision**. See Section 4.7 of Anderson *et al.* (1999) for further details.

8 Further Comments

The real analogue of this routine is F08FCF (DSYEVD).

9 Example

This example computes all the eigenvalues and eigenvectors of the Hermitian matrix A , where

$$A = \begin{pmatrix} 1.0 + 0.0i & 2.0 - 1.0i & 3.0 - 1.0i & 4.0 - 1.0i \\ 2.0 + 1.0i & 2.0 + 0.0i & 3.0 - 2.0i & 4.0 - 2.0i \\ 3.0 + 1.0i & 3.0 + 2.0i & 3.0 + 0.0i & 4.0 - 3.0i \\ 4.0 + 1.0i & 4.0 + 2.0i & 4.0 + 3.0i & 4.0 + 0.0i \end{pmatrix}.$$

The example program for F08FQF (ZHEEVD) illustrates the computation of error bounds for the eigenvalues and eigenvectors.

9.1 Program Text

```
*      F08FQF Example Program Text
*      Mark 20 Revised. NAG Copyright 2001.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LDA
      PARAMETER        (NMAX=8,LDA=NMAX)
      INTEGER          LWORK, LIWORK, LRWORK
      PARAMETER        (LWORK=NMAX*(NMAX+2),LIWORK=5*NMAX+3,
+                      LRWORK=2*NMAX*NMAX+5*NMAX+1)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N
      CHARACTER        JOB, UPLO
*      .. Local Arrays ..
      COMPLEX *16      A(LDA,NMAX), WORK(LWORK)
      DOUBLE PRECISION RWORK(LRWORK), W(NMAX)
      INTEGER          IWORK(LIWORK)
*      .. External Subroutines ..
      EXTERNAL          X04DAF, ZHEEVD
*      .. Executable Statements ..
      WRITE (NOUT,*) 'F08FQF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
        READ (NIN,*) UPLO
*
*      Read A from data file
*
      IF (UPLO.EQ.'U') THEN
        READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
      ELSE IF (UPLO.EQ.'L') THEN
        READ (NIN,*) ((A(I,J),J=1,I),I=1,N)
      END IF
*
      READ (NIN,*) JOB
*
*      Calculate all the eigenvalues and eigenvectors of A
*
      CALL ZHEEVD(JOB,UPLO,N,A,LDA,W,WORK,LWORK,RWORK,LRWORK,IWORK,
+              LIWORK,INFO)
*
      WRITE (NOUT,*)
      IF (INFO.GT.0) THEN
        WRITE (NOUT,*) 'Failure to converge.'
      ELSE
*
*      Print eigenvalues and eigenvectors
*
        WRITE (NOUT,*) 'Eigenvalues'
```

```

                DO 20 I = 1, N
                  WRITE (NOUT,99999) I, W(I)
20             CONTINUE
                WRITE (NOUT,*)
                IFAIL = 0
*
                CALL X04DAF('General',' ',N,N,A,LDA,'Eigenvectors',IFAIL)
*
                END IF
            END IF
*
99999 FORMAT (3X,I5,5X,2F8.4)
        END

```

9.2 Program Data

F08FQF Example Program Data

4		:Value of N
'L'		:Value of UPLO
(1.0, 0.0)		
(2.0, 1.0)	(2.0, 0.0)	
(3.0, 1.0)	(3.0, 2.0)	(3.0, 0.0)
(4.0, 1.0)	(4.0, 2.0)	(4.0, 3.0)
		(4.0, 0.0)
'V'		:End of matrix A
		:Value of JOB

9.3 Program Results

F08FQF Example Program Results

Eigenvalues

1	-4.2443
2	-0.6886
3	1.1412
4	13.7916

Eigenvectors

	1	2	3	4
1	0.4836	0.6470	-0.4456	-0.3859
	0.0000	0.0000	0.0000	0.0000
2	0.2912	-0.4984	-0.0230	-0.4441
	-0.3618	-0.1130	-0.5702	0.0156
3	-0.3163	0.2949	0.5331	-0.5173
	-0.3696	0.3165	0.1317	-0.0844
4	-0.4447	-0.2241	-0.3510	-0.5277
	0.3406	-0.2878	0.2261	-0.3168
